unittest
{
	assert( [ "3", "1 1", "5 1", "5 5" ].parse.expand.solve == 1 );
	assert( [ "2", "10 10", "20 20" ].parse.expand.solve == 0 );
	assert( [ "9", "1 1", "2 1", "3 1", "4 1", "5 1", "1 2", "1 3", "1 4", "1 5" ].parse.expand.solve == 16 );
	
	assert( [ "4", "1 1", "5 1", "1 5", "5 5" ].parse.expand.solve == 0 );		// 追加ケース
}



import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto n = input.front.to!( long );
	auto xys = new long[ 2 ][ n ];
	foreach( ref xy; xys )
	{
		input.popFront;
		xy = input.front.split.to!( long[] );
	}
	return tuple( xys );
}

auto solve( long[ 2 ][] xys )
{
	const XM = 10 ^^ 5;
	const YM = 10 ^^ 5 * 2;
	
	// X座標とY座標で二部グラフを作る
	auto uft = UnionFindTree( YM + 1 );
	foreach( xy; xys )
	{
		uft.unite( xy[ 0 ], XM + xy[ 1 ] );		// X座標とかぶらないようにY座標は下駄を履かせる
	}
	
	// 連結成分ごとにX座標とY座標の種類数を数える
	long[ long ] rxs;
	long[ long ] rys;
	for( auto i = 1L; i <= XM; i++ )
	{
		if( uft.weight( i ) <= 1 ) continue;		// このX座標には頂点が無い
		rxs[ uft.root( i ) ]++;
	}
	for( auto i = XM + 1; i <= YM; i++ )
	{
		if( uft.weight( i ) <= 1 ) continue;		// このY座標には頂点が無い
		rys[ uft.root( i ) ]++;
	}
	
	// 連結成分ごとに完全グラフにしたときの頂点の数を計算
	auto ans = 0L;
	for( auto i = 1L; i <= YM; i++ )
	{
		if( i !in rxs || i !in rys ) continue;
		ans += rxs[ i ] * rys[ i ];
	}
	
	// 完全グラフの頂点数から元の頂点数を抜くと追加する頂点数になる
	return ans - xys.length;
}

// UnionFind木
struct UnionFindTree
{
	private long[] ps;
	private long[] rs;
	
	public this( long n )
	{
		ps.length = n;
		ps[] = -1;
		rs.length = n;
	}
	
	public long root( long i )
	{
		if( ps[ i ] < 0 ) return i;
		else              return ps[ i ] = root( ps[ i ] );
	}
	
	public bool same( long i, long j )
	{
		return root( i ) == root( j );
	}
	
	public void unite( long i, long j )
	{
		i = root( i );
		j = root( j );
		if( i == j ) return;
		if( rs[ i ] < rs[ j ] )
		{
			ps[ j ] += ps[ i ];
			ps[ i ] = j;
		}
		else
		{
			ps[ i ] += ps[ j ];
			ps[ j ] = i;
			if( rs[ i ] == rs[ j ] ) rs[ i ]++;
		}
	}
	
	public long weight( long i )
	{
		return -ps[ root( i ) ];
	}
}
